
Subject: Wrong numbers

Posted by [Mercysh](#) on Fri, 11 Apr 2014 15:52:00 GMT

[View Forum Message](#) <> [Reply to Message](#)

Please help, the totals that I am getting are higher than those published in the reports

Subject: Re: Wrong numbers

Posted by [Erica-DHS](#) on Fri, 11 Apr 2014 17:12:03 GMT

[View Forum Message](#) <> [Reply to Message](#)

Mercy, can you provide some more details? What report and indicators are you working with, and what software package? Please be as specific as possible- once I know more I can direct you to some resources.

Thanks,
Erica

Subject: Re: Wrong numbers

Posted by [Mercysh](#) on Sun, 13 Apr 2014 17:31:21 GMT

[View Forum Message](#) <> [Reply to Message](#)

Thanks Erica

Apologies for not providing details:

As a data quality assessment, I want to calculate age ratios across the population. I ran the table "Household population by age, sex, and residence" for the 2009 Lesotho, 2010 Malawi and Zimbabwe DHS but the numbers I get are higher than those published in the respective countries' reports. I also the de facto population so I do not know what is wrong

Subject: Re: Wrong numbers

Posted by [Mercysh](#) on Sun, 13 Apr 2014 17:31:58 GMT

[View Forum Message](#) <> [Reply to Message](#)

I am using STATA

Subject: Re: Wrong numbers

Posted by [Trevor-DHS](#) on Mon, 14 Apr 2014 01:22:01 GMT

[View Forum Message](#) <> [Reply to Message](#)

Hi Mercy,

We still need more information. Can you provide your code and your output for us to review and see where the difference lies.

Regards. Trevor

Subject: Re: Wrong numbers
Posted by [Mercysh](#) on Mon, 14 Apr 2014 10:54:06 GMT
[View Forum Message](#) <> [Reply to Message](#)

Dear Trevor

I use:

```
generate weight= hv005/1000000  
tab hv105 hv104 [aweight = weight]
```

The results attached are different from that found on page 303 of Lesotho's 2009 report.

File Attachments

1) [age_sex.xlsx](#), downloaded 961 times

Subject: Re: Wrong numbers
Posted by [JanSchuele](#) on Tue, 02 Sep 2014 21:26:19 GMT
[View Forum Message](#) <> [Reply to Message](#)

Hi Mercy,

have you found the cause, that was responsible for the different numbers, in the meantime? I'm interested, because I have the same problem of receiving higher total sample numbers, just for a different country! Would be glad about a short update from your insights.

Kind regards,
Jan

Subject: Re: Wrong numbers
Posted by [Trevor-DHS](#) on Wed, 03 Sep 2014 13:55:18 GMT
[View Forum Message](#) <> [Reply to Message](#)

Hi Mercy,

Your problem is two fold:

- 1) You are not selecting the de facto sample - you need to select on `hv103==1`
- 2) You should use `iweights`, not `aweights`. In fact we actually recommend that you use the `svy`

commands in general, but for a simple crosstab like this iweights will suffice. Other posts in the forum describe the svy commands.

The following commands will produce the results to match the report:

```
generate weight= hv005/1000000  
tab hv105 hv104 if hv103==1 [iweight = weight]
```

Subject: Re: Wrong numbers

Posted by [Mercysh](#) on Thu, 23 Jul 2015 14:12:50 GMT

[View Forum Message](#) <> [Reply to Message](#)

Dear DHS team

I have numbers that are not matching those in the report for Lesotho 2009 but matching for Malawi and Zimbabwe. I want to calculate the % of children <18 living with both parents, father only, mother only and none of the parents. The following is the syntax:

```
generate weight= hv005/1000000  
svyset [pw=weight], psu( hv021) strata(hv022)  
*restrict to usual residence  
keep if hv102==1  
*restrict to <17  
keep if hv105<=17  
*Living arrangements of children  
recode hv114 (0=0 "Not present") (1/30 =1 "Present"), gen(father_presence)  
recode hv112(0=0 "Not present") (1/30 =1 "Present"),gen(mother_presence)  
  
gen parental_presence =.  
replace parental_presence =1 if father_presence == 1 & mother_presence ==1  
replace parental_presence =2 if father_presence == 1 & mother_presence ==0  
replace parental_presence =3 if mother_presence ==1 & father_presence ==0  
replace parental_presence =4 if father_presence == 0 & mother_presence ==0  
  
label var parental_presence parental_presence  
label define parental_presence 1 "Both" 2 "Father only" 3 "Mother only" 4 "None"  
label val parental_presence parental_presence  
svy:tab parental_presence
```

Subject: Re: Wrong numbers

Posted by [Liz-DHS](#) on Wed, 28 Oct 2015 00:21:20 GMT

[View Forum Message](#) <> [Reply to Message](#)

Dear User,

Do you still need assistance with this post?

Thank you!

Subject: Re: Wrong numbers

Posted by [Trevor-DHS](#) on Wed, 28 Oct 2015 14:30:43 GMT

[View Forum Message](#) <> [Reply to Message](#)

Your code does not take into account a few things:

1) The code did not consider whether the father or mother was dead. This is added by checking for . in the recode statements.

2) In Lesotho all family members are listed in the household schedule, including those that do not usually live in the household. To see if they live together you need to check if both the child and the parent usually live in the household. The important piece for this is to check hv102 for the father or mother. This is done with the following logic:

```
hv102[_n-hvidx+hv114] != 1
```

This logic refers to the data record for the father (or mother) by taking the current record number (_n), subtracting the child's line number (hvidx) to find the record before the start of the household, and then adding the father's (or mother's) line number (hv114 or hv112) to find the record for the parent. For this to work, all household members must be kept in the dataset, so the "keep if" commands have been removed, and the selection for children under 18 who were usual members of the household has been put on the svy:tab command instead (if hv102==1 & hv105 <= 17).

3) The code also needs to take into account cases where it was unknown if the father or mother was still alive, that is if hv113 was 8 or 9, or hv111 was 8 or 9.

Try the below code:

```
use LSPR60FL.DTA, clear
```

```
generate weight= hv005/1000000
```

```
svyset [pw=weight], psu(hv021) strata(hv022)
```

```
* sort to ensure that the household listing is in order  
sort hhid hvidx
```

```
*Living arrangements of children
```

```
recode hv114 (0=0 "Not present") (1/30 = 1 "Present") (.=0), gen(father_presence)
```

```
recode hv112 (0=0 "Not present") (1/30 = 1 "Present") (.=0), gen(mother_presence)
```

```
replace father_presence=9 if inrange(hv113,8,9)
```

```
replace mother_presence=9 if inrange(hv111,8,9)
```

```
* need to take into account whether father or mother usually live in the household
```

```
replace father_presence=0 if father_presence==1 & hv102[_n-hvidx+hv114] != 1
```

```
replace mother_presence=0 if mother_presence==1 & hv102[_n-hvidx+hv112] != 1
```

```
*
```

```
gen parental_presence = 9
replace parental_presence = 1 if father_presence == 1 & mother_presence == 1
replace parental_presence = 2 if father_presence == 1 & mother_presence == 0
replace parental_presence = 3 if father_presence == 0 & mother_presence == 1
replace parental_presence = 4 if father_presence == 0 & mother_presence == 0
```

```
label var parental_presence parental_presence
label define parental_presence 1 "Both" 2 "Father only" 3 "Mother only" 4 "None" 9 "DK/Missing"
label val parental_presence parental_presence
svy:tab parental_presence if hv102==1 & hv105 <= 17
```

Subject: Re: Wrong numbers

Posted by [Mercysh](#) on Thu, 06 Jul 2017 16:01:35 GMT

[View Forum Message](#) <> [Reply to Message](#)

I have two follow up questions on the response below:

1. I want to categorise children whose parents are de jure household members, whose parents are migrants (living elsewhere for extended periods), those whose parents are not part of the household because they are dead or are "absent". I am having challenges to do it in stata. I copied your code below for the first two categories but having problems with the rest.

2. Are there other countries besides Lesotho that include individuals who are neither de jure nor de facto members?
